
ЗАДАНИЕ ДЛЯ ВЫПОЛНЕНИЯ РАБОТ 
 
1. Для выполнения каждой лабораторной работы студенту необходимо 

получить код-задание и рассчитать его на TCAD. Построить и отредактировать 
характеристики таким образом, чтобы получить результаты, аналогичные 
приведенным в задании каждой работы. Сохранить полученный результат на 
съемном носителе / в сети (на ПК информацию сохранять не рекомендуется, так как 
есть вероятность её потери). 

2. Построчно и посимвольно разобрать команды, которые используются в коде 
и в комментариях. После кода разместить в отчёте построенные структуры и 
графики с пояснениями элементов, изображенных на них. 

3. Определить, что за структура/прибор были смоделированы, найти описание 
подобных приборов/структур в статьях (типовые геометрические размеры, 
материалы, характеристики) из интернета. Внести информацию в отчёт работы. 
Статьи не должны быть связаны с проектами примеров, приведенных на сайте 
TCAD. 

4. Изменить какой-то параметр в коде в большую и меньшую сторону 
(учитывая адекватность изменений реально существующим технологиям); или 
полностью перестроить структуру исследуемого прибора. Указать, какой параметр 
меняется. Привести структуры и графики прибора с измененными параметрами. 
Объяснить / обосновать результаты внесенных изменений. 

5. Подготовить оформленную работу к отчету. Можно работать и оформлять 
отчеты в парах, защиты проводятся индивидуально. 

 
РАБОТА № 1. МОДЕЛИРОВАНИЕ КНИ СТРУКТУРЫ 

 
Тонкопленочные полностью обедненные МОП-транзисторы со структурой 

КНИ достигают низких пороговых напряжений и почти идеальных обратных 
подпороговых наклонов. Это делает их идеальными для работы с малым 
энергопотреблением. Код задачи показывает, что Atlas способен моделировать как 
транзистор с частично обедненным каналом, так и структуру с полностью 
истощенным каналом. 

Сетка моделирования в этом коде была доработана так, чтобы рассчитывать 
более тонкую структуру КНИ. 

 
Задача 1. PDSOI 
 
go atlas 
TITLE SOI device simulation 

# на SiO2 толщиной 0,4 мкм наносится Si толщиной 0,2 мкм 
 
mesh space.mult=1.0 
 
x.mesh loc=0.00 spac=0.50 
x.mesh loc=1.15 spac=0.02 
x.mesh loc=1.5 spac=0.1 
x.mesh loc=1.85 spac=0.02 
x.mesh loc=3 spac=0.5 
 
y.mesh loc=-0.017 spac=0.02 
y.mesh loc=0.00 spac=0.005 
y.mesh loc=0.1 spac=0.02 
y.mesh loc=0.2 spac=0.01 
y.mesh loc=0.6 spac=0.25 
 
region num=1 y.max=0 oxide 
region num=2 y.min=0 y.max=0.2 silicon 
region num=3 y.min=0.2 oxide 
 
# блок определения электродов (сток, исток, затвор, подложка) 
# #1-GATE  
#2-SOURCE  
#3-DRAIN  
#4-SUBSTRATE (под оксидом) 
 
electrode name=gate x.min=1 x.max=2 y.min=-0.017 y.max=-0.017 
electrode name=source x.max=0.5 y.min=0 y.max=0 
electrode name=drain x.min=2.5 y.min=0 y.max=0 
electrode substrate 
 
# распределение концентрации легирующей примеси в областях 
 
doping uniform conc=2e17 p.type reg=2 
doping gauss n.type conc=1e20 char=0.2 lat.char=0.05 reg=2 x.r=1.0 
doping gauss n.type conc=1e20 char=0.2 lat.char=0.05 reg=2 x.l=2.0 
save outf=project51_0.str 
tonyplot project51_0.str -set project51_0.set 
 
# задание зарядовых областей на обоих границах окисла 

interf qf=3e10 y.max=0.1 
interf qf=1e11 y.min=0.1 
 
# работа выхода затвора 
contact name=gate n.poly 
 
# используемые модели 
models conmob srh auger bgn fldmob print 
 
solve init 
 
# расчёт передаточных ВАХ 
 
method newton trap 
solve prev 
solve vgate=-0.2 
solve vdrain=0.05 
solve vdrain=0.1 
 
# смещение напряжения затвор-исток 
log outf=project51_1.log master 
solve vgate=0.1 vstep=0.1 name=gate vfinal=1.5 
 
# выведение результатов моделирования передаточных ВАХ в стандартный 

редактор 
tonyplot project51_1.log -set project51_1.set 
 
# отображения подпорогового наклона кривых 
tonyplot project51_1.log -set project51_2.set 
 
extract name="subvt" 1.0/slope(maxslope(curve(v."gate",log10(abs(i."drain"))))) 
extract name="vt" (xintercept(maxslope(curve(v."gate",abs(i."drain")))) - 

abs(ave(v."drain"))/2.0) 
quit 
 

РЕЗУЛЬТАТЫ МОДЕЛИРОВАНИЯ 
На рис. 1 и 2 представлены результаты моделирования рассматриваемой 

структуры с требуемыми настройками отображения прибора и его характеристик.  

Рис. 1. Моделирование исходной структуры транзистора 

 

  
а)   б) 

Рис. 2. Передаточные характеристики транзистора (а) с разными настройками 
масштаба для отображения подпорогового наклона (б) 

 
Задача 2. FDSOI 
 
go atlas 
TITLE SOI device simulation 
 
# на подслое из SiO2 толщиной 0,4 мкм нанесен слой Si толщиной 0,1 мкм 
 
mesh space.mult=1.0 
 
x.mesh loc=0.00 spac=0.50 
x.mesh loc=1.15 spac=0.02 

x.mesh loc=1.5 spac=0.1 
x.mesh loc=1.85 spac=0.02 
x.mesh loc=3 spac=0.5 
 
y.mesh loc=-0.017 spac=0.02 
y.mesh loc=0.00 spac=0.005 
y.mesh loc=0.05 spac=0.02 
y.mesh loc=0.1 spac=0.01 
y.mesh loc=0.5 spac=0.25 
 
region num=1 y.max=0 oxide 
region num=2 y.min=0 y.max=0.1 silicon 
region num=3 y.min=0.1 oxide 
 
electrode name=gate x.min=1 x.max=2 y.min=-0.017 y.max=-0.017 
electrode name=source x.max=0.5 y.min=0 y.max=0 
electrode name=drain x.min=2.5 y.min=0 y.max=0 
electrode substrate 
 
doping uniform conc=1e17 p.type reg=2 
doping gauss n.type conc=1e20 char=0.2 lat.char=0.05 reg=2 x.r=1.0 
doping gauss n.type conc=1e20 char=0.2 lat.char=0.05 reg=2 x.l=2.0 
save outf=project52_0.str 
tonyplot project52_0.str -set project52_0.set 
interf qf=3e10 y.max=0.05 
interf qf=1e11 y.min=0.05 
 
contact name=gate n.poly 
models conmob srh auger bgn fldmob print 
 
solve init 
 
method newton trap 
solve prev 
solve vgate=-0.2 
solve vdrain=0.05 
solve vdrain=0.1 
 
log outf=project52_1.log master 
solve vgate=0.1 vstep=0.1 name=gate vfinal=1.5 

tonyplot project52_1.log -set project52_1.set 
tonyplot project52_1.log -set project52_2.set 
 
extract name="subvt" 1.0/slope(maxslope(curve(v."gate",log10(abs(i."drain"))))) 
extract name="vt" (xintercept(maxslope(curve(v."gate",abs(i."drain")))) - 

abs(ave(v."drain"))/2.0) 
 
quit 
 

РЕЗУЛЬТАТЫ МОДЕЛИРОВАНИЯ 
На рис. 3 и 4 представлены результаты моделирования рассматриваемой 

структуры с требуемыми настройками отображения прибора и его характеристик.  
 

 
 

Рис. 3. Моделирование исходной структуры транзистора 

Рис. 4. Передаточные характеристики транзистора (а) с разными настройками 

масштаба для отображения подпорогового наклона (б) 
 

РАБОТА № 2. МОДЕЛИРОВАНИЕ ТРЕХМЕРНОЙ КНИ СТРУКТУРЫ 
 
В этом коде демонстрируется моделирование 3D КНИ структуры в модуле 

Atlas. Конкретно это моделирование не может быть выполнено с помощью 2D-
симулятора, поскольку корпусной электрод расположен вне плоскости стока, 
затвора и истока. 

 
go atlas 
 
TITLE 3D simulation of a Body Contacted SOI MOSFET 
 
# на пленку SiO2 толщиной 0,4 мкм наносится слой кремния толщиной 0,05 

мкм 
 
mesh three.d space.mult=1.0 
 
x.mesh loc=0.00 spac=0.5 
x.mesh loc=2.0 spac=0.5 
x.mesh loc=2.8 spac=0.1 
x.mesh loc=2.95 spac=0.02 
x.mesh loc=3.2 spac=0.2 
x.mesh loc=3.45 spac=0.02 
x.mesh loc=3.6 spac=0.1 
x.mesh loc=4.50 spac=0.5 
x.mesh loc=6.5 spac=0.5 

y.mesh loc=-0.022 spac=0.02 
y.mesh loc=-0.012 spac=0.02 
y.mesh loc=0.00 spac=0.0025 
y.mesh loc=0.025 spac=0.005 
y.mesh loc=0.05 spac=0.005 
y.mesh loc=0.1 spac=0.05 
y.mesh loc=0.5 spac=0.2 
 
z.mesh loc=0 spac=0.5 
z.mesh loc=1 spac=0.5 
z.mesh loc=2 spac=0.5 
z.mesh loc=2.5 spac=0.1 
z.mesh loc=3.5 spac=0.25 
z.mesh loc=4 spac=0.1 
 
region num=1 y.max=0 z.min=0 z.max=4 oxide 
region num=2 y.min=0 y.max=0.05 z.min=0 z.max=4 silicon 
region num=3 y.min=0.05 z.min=0 z.max=4 oxide 
 
electrode name=source x.max=2.0 y.min=-0.012 y.max=0 z.min=0 z.max=1 
electrode name=gate x.min=2.8 x.max=3.6 y.min=-0.022 y.max=-0.012 z.min=0 

z.max=2.5 
electrode name=drain x.min=4.5 y.min=-0.012 y.max=0 z.min=0 z.max=1 
electrode name=body x.min=2.8 x.max=3.6 y.min=-0.012 y.max=0.0 z.min=3.5 

z.max=4 
electrode substrate 
 
doping uniform p.type conc=1.5e17 
doping uniform n.type conc=5.e19 x.left=0. x.right=3.0 y.top=0.0 y.bottom=0.05 

z.min=0 z.max=2 
doping uniform n.type conc=5.e19 x.left=3.5 x.right=6.5 y.top=0.0 y.bottom=0.05 

z.min=0 z.max=2 
doping uniform p.type conc=5.e19 x.left=2.8 x.right=3.6 y.top=0.0 y.bottom=0.05 

z.min=3.5 z.max=4 
 
save outf=project6_0.str 
 
contact name=gate n.poly 
models conmob srh auger bgn fldmob print 
impact selb 

tonyplot3d project6_0.str -set project6_0.set 
solve init 
solve prev 
 
method bicgst maxtrap=6 trap 
solve vdrain=0.025 
solve vdrain=0.05 
solve vdrain=0.1 
 
log outf=project6_1.log 
 
solve vgate=0.025 vstep=0.025 vfinal=0.2 name=gate 
solve vgate=0.3 vstep=0.1 vfinal=1.5 name=gate 
 
tonyplot project6_1.log -set project6_1.set 
 
# выходная ВАХ 
# смещаем электрод затвора до 1,5 В 
 
log off 
solve init 
 
method bicgst maxtrap=6 trap 
 
solve vgate=0.5 
solve vgate=1 
solve vgate=1.5 
 
# смещаем электрод стока и сохраняем результирующую ВАХ в файл 
 
log outf=project6_2.log master 
solve vdrain=0.00625 
solve vdrain=0.0125 
solve vdrain=0.025 
solve vdrain=0.05 
solve vdrain=0.1 vstep=0.1 vfinal=1 name=drain 
solve vdrain=1.25 vstep=0.25 vfinal=3 name=drain 
 
method bicgst maxtrap=6 trap 
solve vdrain=3.25 vstep=0.25 name=drain vfinal=4 

tonyplot project6_2.log -set project6_2.set 
tonyplot project6_2.log -set project6_3.set 
quit 
 

РЕЗУЛЬТАТЫ МОДЕЛИРОВАНИЯ 
На рис. 5 и 6 представлены результаты моделирования рассматриваемой 

структуры с требуемыми настройками отображения прибора и его характеристик.  
 

 
Рис. 5. Исходная структура трехмерного КНИ тонкопленочного транзистора 

 

   
Рис. 6. Графики функций, характерных для данного транзистора  

(виды графиков студентам требуется определить самостоятельно) 

РАБОТА № 3. МОДЕЛИРОВАНИЕ ТЕХНОЛОГИЧЕСКОГО ПРОЦЕССА 
ИЗГОТОВЛЕНИЯ КНИ СТРУКТУРЫ 

 
В коде работы реализуется моделирование процесса разработки КНИ МОП-

транзистора с полной диэлектрической проницаемостью. Затем выполняется 
моделирование характеристик этого устройства с использованием модели 
неизотермического баланса энергии (NEB). Код моделирования технологического 
процесса и выходных характеристик описывает: 

– формирование структуры КНИ с использованием технологического 
процесса SSuprem 4; 

– частично обедненную структуру КНИ; 
– выбор моделей энергобаланса и разогрева кристаллической решетки в 

модуле Atlas; 
– выбор численных методов для метода парных решений; 
– моделирование выходных ВАХ при значениях Vgs = 1.0, 3.0 и 5.0 В. 
Структура кремний-оксид-кремний была сформирована в SSuprem 4 с 

использованием определения областей в редакторе Athena. Затем эта структура была 
подвергнута стандартной последовательности обработки объемного MOSFET. 
Структура была спроектирована таким образом, чтобы толщина КНИ была больше 
глубины стыков истока и стока. Это гарантирует, что устройство будет работать в 
частично обедненном режиме при последующем моделировании работы прибора. 

Одним из основных изменений, внесенных в стандартную последовательность 
изготовления МОП-транзистора, было введение LOCOS – этапа создания особого 
слоя диэлектрика (студенты, найдите информацию по данной технологии!). 
Результатом использования данного этапа технологического процесса будет 
создание полностью изолированного окислом островка кремния.  

В данном коде для окисления используется модель COMPRESS, так как она 
более сложная и точная. Модель может использоваться для изучения эффектов 
«птичьего клюва» в полевых МОП-транзисторах с КНИ, информация о котором есть 
в мануалах по модулю Athena и описанию технологического процесса SSuprem 4. 

Модель технологического процесса автоматически передается в модуль Atlas, 
в котором происходит моделирование ВАХ созданного прибора.  

 
go athena 
 
TITLE: SOI Process and Device Simulation Example 
 
# Задание сетки подложки 
line y loc=0 spac=0.02 tag=top 
line y loc=0.13 spac=0.01 

line y loc=0.2 spac=0.02 tag=oxtop 
line y loc=0.7 spac=0.5 
line y loc=1.2 spac=0.1 tag=oxbot 
line y loc=2 spac=0.5 tag=bot 
 
line x loc=-1 spac=0.5 tag=left 
line x loc=0 spac=0.1 
line x loc=1.55 spac=0.01 
line x loc=2 spac=0.25 tag=right 
 
region silicon xlo=left xhi=right ylo=top yhi=oxtop 
region oxide xlo=left xhi=right ylo=oxtop yhi=oxbot 
region silicon xlo=left xhi=right ylo=oxbot yhi=bot 
 
init orient=100 c.phos=1e14 space.mult=1.5 
 
# нанесение окисла и нитрида кремния 
deposit oxide thick=0.02 div=1 
deposit nitride thick=0.1 div=1 
 
# с данного этапа моделирование отображается в 2D режиме 
 
etch nitride left p1.x=0 
etch oxide left p1.x=0 
 
# окисление с определенным режимом технологического процесса  
 
method compress fermi 
diffus time=90 temp=1000 weto2 press=1.00 hcl.pc=0 
 
etch nitride all 
 
etch oxide start x=1 y=-1 
etch continue x=3 y=-1 
etch continue x=3 y=0.1 
etch done x=1 y=0.1 
 
# формирование p-кармана и скрытого n-кармана 
 
diffus time=30 temp=1000 dryo2 press=1.00 hcl=3 

etch oxide thick=0.02 
 
# имплантация p-кармана 
implant boron dose=8e12 energy=100 pears 
 
# имплантация скрытого n-кармана (в данном коде не отображается) 
 
# термические процессы разгонки примеси / отжига структуры  
diffus time=220 temp=1200 nitro press=1 
diffus time=90 temp=1200 t.rate=-4.444 nitro press=1 
etch oxide thickness=0.03 
 
# с данного этапа происходит создание области затвора (первый этап – 

окисление) 
diffus time=11 temp=925 dryo2 press=1.00 hcl=3 
 
# извлечение параметров рассчитываемой структуры 
 
extract name="gateox" thickness oxide mat.occno=1 x.val=2.0 
 
# задание порогового напряжения имплантацией бора  
implant boron dose=9.5e11 energy=10 pearson 
 
# наращивание поликремниевого затвора 
depo poly thick=0.2 divi=2 c.phosphor=2e20 
etch poly left p1.x=1.5 
 
# окисление поликремния 
method fermi compress 
diffuse time=3 temp=900 weto2 press=1.0 
 
# имплантация LDD n-типа 
implant phosphor dose=3.0e13 energy=20 pearson 
 
# создание боковых стенок 
depo oxide thick=0.120 divisions=8 
etch oxide dry thick=0.120 
 
# имплантация областей истока/стока 
implant arsenic dose=5.0e15 energy=50 pearson 

method fermi compress 
diffuse time=2 temp=900 nitro press=1.0 
etch oxide thickness=0.02 
 
# нанесение металла и вытравливание рельефа контактов S/G/D 
deposit alumin thick=0.03 divi=2 
 
struct mirror right 
 
etch alumin right p1.x=4.5 
etch alumin left p1.x=-0.5 
 
etch alumin start x=1.25 y=-1 
etch continue x=2.75 y=-1 
etch continue x=2.75 y=0.1 
etch done x=1.25 y=0.1 
 
# определение электродов структуры 
electrode name=gate x=2.0 y=-0.1 
electrode name=source x=0.0 
electrode name=drain x=4.0 
electrode name=substrate backside 
 
# сохранение структуры в файл 
structure outfile=project7_0.str 
tonyplot project7_0.str -set project7_0.set 
 
go atlas 
 
# задание работы выхода электрода затвора 
contact name=gate n.poly 
 
# задание заряда Qss на границе окисла с затвором  
interface qf=3e10 
 
# подключение модели CVT, используемой при моделировании МОП-

структуры 
models cvt srh hcte.el lat.temp 
impact selb 

thermcontact num=1 elec.num=4 ext.temp=300 
 
#инициализация процесса моделирования 
method newton trap autonr maxtrap=10 
 
solve init 
 
output e.field j.electron j.hole j.conduc j.total ex.field ey.field \ 
flowlines e.mobility h.mobility qss e.temp h.temp val.band con.band \ 
qfn qfp j.disp photogen impact 
 
# приложение малого начального смещения стока 
solve vfinal=0.25 vstep=0.05 name=drain 
 
# расчёт ВАХ при Vg=1, 3, 5 В 
solve vfinal=1 vstep=0.25 name=gate 
save outf=project7_1.str 
 
solve vfinal=3 vstep=0.25 name=gate 
save outf=project7_3.str 
 
solve vfinal=5 vstep=0.25 name=gate 
save outf=project7_5.str 
 
# расчёт выходной ВАХ при Vg=1 В 
load infile=project7_1.str master 
solve prev 
log outf=project7_1.log 
solve vfinal=5 vstep=0.25 name=drain 
save outf=project7_1a.str 
 
log outf=tmp 
 
# расчёт выходной ВАХ при Vg=3 В 
load infile=project7_3.str master 
solve prev 
log outf=project7_3.log 
solve vfinal=5 vstep=0.25 name=drain 
save outf=project7_3a.str 
log outf=tmp 

# расчёт выходной ВАХ при Vg=5 В 
load infile=project7_5.str master 
solve prev 
log outf=project7_5.log 
solve vfinal=5 vstep=0.25 name=drain 
save outf=project7_5a.str 
 
tonyplot -overlay project7_1.log project7_3.log project7_5.log -set project7_1.set 
tonyplot -overlay project7_1.log project7_3.log project7_5.log -set project7_2.set 
tonyplot project7_5a.str -set project7_5a.set 
quit 
 

РЕЗУЛЬТАТЫ МОДЕЛИРОВАНИЯ 
На рис. 7 и 8 представлены результаты моделирования рассматриваемой 

структуры с требуемыми настройками отображения прибора и его характеристик.  
 

    
Рис. 7. Структура смоделированного в модуле Athena транзистора 

 

   
Рис. 8. Выходная ВАХ и температурная зависимость ВАХ транзистора 

РАБОТА № 4. МОДЕЛИРОВАНИЕ СОЛНЕЧНОГО ЭЛЕМЕНТА  
С НАНОРАЗМЕРНЫМИ ОБЛАСТЯМИ И ИССЛЕДОВАНИЕ                        

ЕГО ХАРАКТЕРИСТИК 
 
В данной работе солнечный элемент (СЭ) реализован в виде коаксиального 

кремниевого наношнура толщиной порядка 380 нм. Предполагается, что наношнур 
состоит из трех слоев кремния – положительно заряженного ядра, тонкого нейтрально 
заряженного или с собственным зарядом промежуточного слоя и отрицательно 
заряженного внешнего слоя. Когда фотон ударяет в наношнур, он генерирует 
электронно-дырочную пару заряда. Заряд разделяется и движется радиально: 
электроны направляются к внешнему электроду, а дырки – к внутреннему. 

Преимуществом подобной структуры по сравнению с классическими СЭ 
является более короткое расстояние, которое проходят электрон и дырка до 
выходного контакта СЭ. 

В модуле Atlas3D параметр CYLINDRICAL задает саму конструкцию 
цилиндрической структуры. Сетка устанавливается как MESH THREE.D 
CYLINDRICAL, где параметр THREE.D указывает симулятору на необходимость 
создания полноценной трехмерной сетки. В этом случае требуется указывать 
настройки будущего прибора – радиус шнура, углы шага и длину в декартовых 
координатах. 

 
go atlas 
 
mesh cylindrical three.d 
 
r.mesh l=0 spacing=0.05 
r.mesh l=0.08 spacing=0.01 
r.mesh l=0.16 spacing=0.05 
r.mesh l=0.19 spacing=0.01 
 
# распределение сечения по углам и шаг углов 
a.m l=0 spac=60 
a.m l=360 spac=60 
 
# длина шнура 
z.mesh l=0 spacing=0.01 
z.mesh l=0.03 spacing=0.01 
z.mesh l=0.11 spacing=0.025 
z.mesh l=3.105 spacing=1 
z.mesh l=3.11 spacing=0.0015 

region num=1 material=Silicon a.min=0 a.max=360 z.min=0 z.max=0.03 r.min=0 
r.max=0.19 

region num=2 material=Silicon a.min=0 a.max=360 z.min=0.03 z.max=3.105 
r.min=0.16 r.max=0.19 

region num=3 material=Silicon a.min=0 a.max=360 z.min=0.03 z.max=0.11 
r.min=0 r.max=0.16 

region num=4 material=Silicon a.min=0 a.max=360 z.min=0.11 z.max=3.105 
r.min=0.08 r.max=0.16 

region num=5 material=Silicon a.min=0 a.max=360 z.min=0.11 z.max=3.11 
r.min=0 r.max=0.08 

region num=6 material=air a.min=0 a.max=360 z.min=3.105 z.max=3.11 
r.min=0.08 r.max=0.29 

 
electrode name=cathode a.min=0 a.max=360 r.min=0 r.max=0.19 z.min=0 z.max=0 
electrode name=anode a.min=0 a.max=360 r.min=0 r.max=0.19 z.min=3.11 

z.max=3.11 
 
doping region=1 uniform n.type conc=5e+17 
doping region=2 uniform n.type conc=5e+17 
doping region=3 uniform n.type conc=1e+14 
doping region=4 uniform n.type conc=1e+14 
doping region=5 uniform p.type conc=4e+16 
 
save outf=project8_0.str 
 
beam num=1 z.origin=-2.0 raytrace=project8_1.str x.origin=0 y.origin=0 phi=0.0 

theta=90 nx=10 nz=10 power.file=solar.spec 
 
models fermi ni.fermi cvt srh 
 
method BICGST 
 
solve init 
log outf=project8_0.log 
solve previous 
solve vstep=0.02 vfinal=0.55 name=anode 
log off 
solve init 
solve b1=1 
log outf=project8_1.log 

solve previous 
solve vstep=0.02 vfinal=0.55 name=anode 
 
tonyplot -overlay project8_0.log project8_1.log -set project8_0.set 
tonyplot3d project8_1.str -set project8_1.set 
 
extract init inf="project8_1.log" 
extract name="Jsc" y.val from curve(v."anode", i."cathode") where x.val=0.0 
extract name="Voc" x.val from curve(v."anode", i."cathode") where y.val=0.0 
extract name="Pm" max(curve(v."anode", (v."anode" * i."cathode"))) 
extract name="Vm" x.val from curve(v."anode", (v."anode"*i."cathode") ) where 

y.val=$"Pm" 
extract name="Im" $"Pm"/$"Vm" 
extract name="FF" ($"Pm"/($"Jsc"*$"Voc")) 
extract name="Eff" ($"Pm"/(3.14157*(0.19)*(0.19)*1e-8*0.1))*100 
quit  
 

РЕЗУЛЬТАТЫ МОДЕЛИРОВАНИЯ 
На рис. 9 и 10 представлены результаты моделирования рассматриваемой 

структуры с требуемыми настройками отображения прибора и его характеристик.  
 

 
Рис. 9. Структура СЭ в 3D 

Рис. 10. ВАХ СЭ при освещении и без него 

